DESIGNED FOR HP-IB SYSTEMS

# Programming Note

8566A/8568A/9835/9845-1 NOVEMBER 1980

SUPERCEDES: NONE

# Introductory Operating Guide
## for the 8566A/8568A Spectrum Analyzers with the 9835/9845 Desktop Computers



HEWLETT PACKARD

## Introduction

This note is an introductory guide to remote operation and programming of 8566A and 8568A Spectrum Analyzers using either the 9835 or 9845 Desktop Computer. Included in this guide are system connections for remote operation and several example programs with descriptions of each step.

The 8566A and 8568A are microprocessor-controlled, general purpose spectrum analyzers which are compatible with the Hewlett-Packard Interface Bus (HP-IB). When used with any HP-IB controller, such as the 9835 or 9845, these instruments become fully automated spectrum analyzers featuring:

* Precise, stable LO tuning
* High sensitivity and resolution
* Wide dynamic range

## Related Documents

Complete operating information for the 8566A/8568A analyzers can be found in:

1. 8566A/8568A Spectrum Analyzer Operation
   (P/N 08566-90002 or 08568-90002)
2. 8566A/8568A Spectrum Analyzer Remote Operation
   (P/N 08566-90003 or 08568-90003)
3. 8566A/8568A Spectrum Analyzer Pull-Out Information Cards

Information on operating the 9835 and 9845 controllers can be found in:

1. System 35/45 Operating and Programming Manual
2. System 35/45 Beginner's Guide
3. System 35/45 Reference Guide
4. System 35/45 I/O ROM Programming

A description of interface programming and hardware can be found in:

1. BASIC Language Interfacing Concepts
   (P/N 09835-90600)

## Equipment Required

To perform the examples in this note, you will need the following equipment and accessories:

1. 8566A or 8568A Spectrum Analyzer
2. 9835A/B Desktop Computer with 98332A I/O ROM, or
   9845B/T Desktop Computer with 98412A I/O ROM (Option 312)
3. 98034A HP-IB Interface*

## Setup

Figure 1 shows the system connections and switch setting for the 98034A HP-IB Interface. To connect the system as shown, follow these steps:

*revised cards only

1. Turn off power to the 9835/9845.
2. Install the I/O ROM in any available socket (in front on the 9835, or in the left side drawer on the 9845).
3. Install the 98034A in any available socket on the rear of the 9835/9845. Be sure the 98034A seats securely in the socket; this has occurred when the latch on top of the interface pops up, locking the card into the socket.
4. Set the rotary switch located on top of the 98034A to position 7. Seven is the select code of the interface for all programs in this guide.
5. Connect the 24-pin connector at the free end of the 98034A cable to the rear panel of the 8566A/8568A (see Figure 1). The connector is shaped to ensure proper orientation.

**CAUTION**

**Do not attempt to mate silver English threaded screws on one connector with black metric threaded nuts on another connector, or vice versa, as damage to the hardware may result. A metric conversion kit which will convert one cable and one or two instruments to metric hardware may be obtained by ordering HP P/N 5060-0138.**




*Figure 1. System connection.*

## Check-Out

After making AC power line connections to the analyzer, the STANDBY lights on both the RF and display sections should be illuminated. Switch on the 9835/9845 and set the analyzer to LINE ON.




LINE ON and STANDBY




INSTR CHECK LED

Upon LINE ON, the analyzer will perform an automatic internal instrument check, designated by the red INSTR CHECK indicators. Both LED's will turn on momentarily during the brief check routine and, if the instrument is operating properly, will go off and remain off during operation, except when another instrument check is triggered by an Instrument Preset. If one or both LED's remain on, refer to the 8566A/8568A Operating and Service Manual, Section II.

Verify that the analyzer's address is set to 18. The read/write address of the 8566A or 8568A can be determined and altered from the front panel by using the shift function P:

Pressing 


sets the address to 18.

When the analyzer is turned on from a cold state, crt messages OVEN COLD and REF UNLOCK may appear. These will go off typically ten minutes after AC power is connected. Type the following commands on the controller keyboard:

`ABORTIO 7` (Press EXECUTE)
`REMOTE 718` (Press EXECUTE).

If ADRS'D and REM light up on the analyzer's front panel, proceed to the programming examples. If either ADRS'D or REM do not light, check to make sure that the 98034A select code is set to 7, the interface cables are properly connected, and the address in the REMOTE statement matches the address of the 8566A/8568A. Although 18 is the factory-set address and the address used in the following examples, other addresses are possible.

If both ADRS'D and REM still do not light, consult the 8566A/8568A Operating and Service Manual, the System 35/45 Test Manual, and the 98034A Installation and Service Manual for troubleshooting information.

## Programming Examples

The following examples illustrate some of the ways to operate the 8566A/8568A using the 9835/9845 controller.

The examples illustrate setting front panel controls remotely and outputting their values, outputting marker values, and outputting trace data. An example harmonic distortion measurement program incorporates some of these techniques in a typical application.

### EXAMPLE 1: PROGRAMMING FRONT-PANEL FUNCTIONS

To preset the analyzer, and set center frequency to 100 MHz and span to 10 MHz, enter the following on the keyboard of the 9835/9845 controller: `OUTPUT 718;"IP CF100MZ SP10MZ"`

```
OUTPUT   7   18;   "IP   CF   100MZ      SP   10MZ"
```

- `OUTPUT` — controller TALK
- `7` — select HP-IB
- `18;` — analyzer address
- `"IP` — instrument preset
- `CF` — activate center frequency
- `100MZ` — set center frequency value
- (space) — optional spaces inserted for clarity
- `SP` — activate frequency span
- `10MZ"` — set frequency span value

Executing this statement initiates the sequence of operations shown above. The final crt display with a 100 MHz signal present should look like this:




The last function activated, SPAN, will appear with its current value on the analyzer crt as shown in the shaded box.

## NOTE

**An important concept in analyzer programming is worthy of special note here. The sequence of operations executed above could have been entered manually from the front panel of the analyzer to yield the same result. In fact, a manual sequence of keystrokes is usually developed first and then used as a basis for executing the same procedure under program control. This simple technique is recommended as a powerful tool for software development with the automatic spectrum analyzer.**

## EXAMPLE 2: OUTPUTTING A FUNCTION OR MARKER VALUE

In the first case, a BASIC program is shown which directs the analyzer to activate center frequency, and to prepare to output the current value in a subsequent statement. The value is then transferred into the variable F and printed. The END statement, line 50, terminates the program.

```
10    OUTPUT 718;"CF OA"                     ! Activate center frequency, prepare
20                                           ! to output value of active function.
30    ENTER 718;F                            ! Transfer value to F.
40    PRINT "Center Frequency =";F;"Hz"      ! Print value.
50    END
```

To enter the program, press:

EDIT [Press EXECUTE].

```
10 _
```

should appear. Type a line and press STORE. Now

```
20 _
```

should appear. Continue entering program code line by line. After storing the last line, END, press RUN to execute the program.* (Omit annotation which begins with "!" on each line, or entire lines which contain only annotation; these comments are provided for the reader's clarification only. Note that your line numbers will not in general correspond to those in this guide.)

A typical output would be:

```
Center Frequency = 100000000 Hz
```

Next, we would like to output both the amplitude and frequency of the active marker. To illustrate this, connect the analyzer's CAL OUTPUT to the RF INPUT. Type SCRATCH A and press EXECUTE to clear the program memory, and enter the following program:

```
10    OUTPUT 718;"IP FA75MZ FB150MZ S2 TS E1"
20                            ! Instrument preset, set start and stop freq's,
30                            ! single sweep, take sweep, peak search.
40    OUTPUT 718;"MA"                   ! Prepare to output marker amplitude.
50    ENTER 718;A                       ! Transfer amplitude into variable A.
60    OUTPUT 718;"MF"                   ! Prepare to output marker frequency.
70    ENTER 718;F                       ! Transfer frequency into variable F.
80    PRINT A;"dBm   ";F/1E6;"MHz"      ! Print A and F (scaled to megahertz).
90    END
```

*For a brief introduction to the controller editing facilities, refer to the Editing Section of the chapter entitled Keyboard Operations in the System 35/45 Operating and Programming Manual.

The first line presets the analyzer, sets start and stop frequencies to 75 MHz and 150 MHz, and then instructs the analyzer to use the single sweep mode. To ensure that a trace is displayed which corresponds to the current instrument control settings, a take sweep command ("TS") is used. This triggers a sweep and prevents the analyzer from accepting further commands until the trace is complete.

Upon completion of this sweep, the peak search ("E1") command is invoked, placing a marker on the largest signal displayed. Lines 40 and 50 instruct the analyzer to output the amplitude value in dBm into the variable A, and lines 60 and 70 cause the frequency value in hertz to be transferred into F. These two values are then printed with appropriate units. Note that the frequency in hertz has been divided by one million to yield megahertz.

Pressing RUN yields typical output:

```
-10.4 dBm     100.2 MHz
```

## EXAMPLE 3: OUTPUT TRACE DATA

An important capability of an automatic spectrum analyzer is to transfer trace amplitude data into an array in the controller for subsequent manipulation. A direct approach is shown in the first program:

```
10   DIM A(1000)                 ! Dimension array A from 0 to 1000
20                               ! (1001 points total).
30   OUTPUT 718;"S2 TS O3 TA"    ! Using O3 format (reference level units),
40                               ! prepare to output trace A.
50   FOR N=0 TO 1000             ! Begin FOR-NEXT loop.
60   ENTER 718;A(N)              ! Transfer formatted data one point at a time
70                               ! into A array.
80   NEXT N                      ! End of FOR-NEXT loop.
90   FOR N=0 TO 1000 STEP 100      !
100  PRINT N,A(N)                  ! Print every one-hundredth point
110  NEXT N                        !
120  END
```

After dimensioning the array, four commands are sent to the analyzer in the OUTPUT 718 statement. First, the analyzer is set to the single sweep mode, followed by a take sweep command. The single sweep mode ("S2") is especially important when outputting trace data because it provides a static display while the values are being accessed. Following the TS command (discussed in Example 2) there is an output format command O3. (This is the letter O for Output, not zero!) The analyzer in this mode scales the display units from the ADC (analog-to-digital converter) to reference level units (in this example, dBm), and re-formats these values into a sequence of ASCII characters which will be transmitted over the interface bus. TA specifies trace A data, which are subsequently transferred one point at a time into the A array using the ENTER 718 statement repeated 1001 times.

Finally, to show what has happened, several data values are printed.

```
0                    -91.9
100                  -85.2
200                  -86.4
300                  -85.5
400                  -82.1
500                  -10.9
600                  -77
700                  -81.8
800                  -87.8
900                  -84.3
1000                 -87
```

The running time for this program using a 9835 is about 26 seconds. To achieve a faster transfer, we can avoid rescaling the ADC values and re-formatting into ASCII code by using O2 instead of O3 output format. We can then achieve a very efficient means of transferring the trace data as unformatted binary values through the use of a byte-by-byte fast handshake command.

In the case below, a sequence of 8-bit bytes is transferred into the integer-valued A array. Note that the values in the A array are two bytes or sixteen bits long, as are the binary values to be transferred from the analyzer in the O2 format mode. Therefore, we have specified that 2002 bytes be transferred, which corresponds exactly to 1001 values. The values which are printed from the A array are in display units. These range from 0 to 1023, and may be accessed as such for further processing. A typical execution time for this transfer using the 9835/9845 is 200 milliseconds.

```
10    INTEGER A(1000)                              ! Dimension A array from 0 to 1000
20                                                 ! (1001 points total).
30    OUTPUT 718;"S2 TS O2 TA"             ! Single sweep, take sweep, using format O2
40                                         ! (binary units) prepare to output trace A.
50    ENTER 718 BFHS 2002 NOFORMAT;A(*)      ! Transfer data into array using byte-
60                                           ! by-byte fast handshake, no format.
70    FOR N=0 TO 1000 STEP 100             !
80    PRINT N,A(N)                         ! Print every one-hundredth point.
90    NEXT N                               !
100   END
```

| | |
|---|---|
| 0 | 108 |
| 100 | 133 |
| 200 | 126 |
| 300 | 124 |
| 400 | 186 |
| 500 | 890 |
| 600 | 249 |
| 700 | 119 |
| 800 | 153 |
| 900 | 149 |
| 1000 | 125 |

This program illustrates how more advanced BASIC programming techniques can be implemented to produce significantly higher performance in the area of automatic instrument control. Such topics as advanced transfer techniques are treated in the System 35/45 I/O ROM Programming manuals.

## NOTE

**Correct format usage when transferring data and commands to and from the analyzer is essential for proper operation under remote control. Errors in formatting are a frequent cause of program failure; study the format codes if you are not certain of correct usage when debugging a program under development.**

**Data are transferred over the interface bus one 8-bit byte at a time. These may be ASCII-encoded alphanumeric characters, or binary values. For example, when the O3 format has been specified (this is the default mode on instrument preset) and a trace value is output from the analyzer, a sequence of ASCII characters is transmitted across the bus, as many as needed to specify the value of interest. The analyzer automatically performs the necessary formatting from an internally stored binary value to an ASCII string, and the controller reverses this process on receipt of such a string. As the number of characters transferred is variable, a free field format is required in the control program.**

**Alternatively, data values themselves may be transferred in 8-bit bytes (two bytes will be necessary to retain the full 10-bit precision of values stored in the analyzer). Here, the analyzer may be in the O2 format, and the controller in an unformatted or binary formatted mode (i.e., ASCII formatting must not occur). This is illustrated in the second trace output example involving the byte-by-byte fast hand shake transfer mode.**

**See the Spectrum Analyzer Remote Operation manual for further information on input/output formats.**

## EXAMPLE 4: HARMONIC DISTORTION MEASUREMENT

An example program which illustrates some of the techniques demonstrated above is included here. This program makes a harmonic distortion measurement by locating and measuring a signal's second and third harmonics and calculating the percent distortion relative to the fundamental. The technique suggested in Example 1 - converting a manual sequence of keystrokes into a program to perform the same functions - was used in developing the present example.

```
10   ! HARMONIC DISTORTION MEASUREMENT
20   ! REV A, 801024
30   !
40   OUTPUT 718;"IP"
50   LOCAL 718
60   DISP "Set analyzer to display the fundamental signal."
70   PAUSE
80   DISP ""
90   OUTPUT 718;"SP 03 OA"          ! Prepare to output the current span.
100  ENTER 718;Span                 ! Transfer value (in hertz) to "Span".
110  Span=MIN(Span,1E5)             ! Use current value or 100 kHz,
120                                 ! whichever is smaller.
130  OUTPUT 718;"S2 TS E1 MT1 SP",Span,"HZ TS MT0 E4 TS E1 E3 MA"
140            ! Acquire signal with peak search, auto-zoom, marker to reference
150            ! level, peak search; enter CF STEP SIZE with E3 command; use MA
160            ! to prepare to output fundamental amplitude.
170  ENTER 718;Fund                 ! Transfer marker amplitude to "Fund".
180  OUTPUT 718;"MF"                ! Prepare to output marker frequency.
190  ENTER 718;Freq                 ! Transfer marker freq to "Freq"
200  Freq=Freq/1E6                  ! Scale frequency to megahertz.
210  OUTPUT 718;"CF UP TS E1 MA"    ! Increment center freq by fundamental freq.
220  ENTER 718;Second               ! Transfer marker amplitude to "Second".
230  OUTPUT 718;"CF UP TS E1 MA"    ! Increment center freq by fundamental freq.
240  ENTER 718;Third                ! Transfer marker amplitude to "Third".
250  Dist=100*SQR(FNLin(Second)^2+FNLin(Third)^2)/FNLin(Fund)
260                                 ! Compute root-sum-of-the-squares
265                                 ! total harmonic distortion using "Lin"
270                                 ! function defined below.
280                                 !
290 Format1:  IMAGE 4A,XSDDD.DX,"dBm",XXXKX,"MHz" !____
300  PRINT USING Format1;"Fund",Fund,Freq           !____     Formatted
310 Format2:  IMAGE 2(4A,XSDDD.DX,"dBm",/)            !____    output.
320  PRINT USING Format2;"2nd ",Second,"3rd ",Third       !____
330  PRINT USING "K,DDD.DD,K//";"Harmonic Distortion = ";Dist;"%"  !
340                                 !
350  DEF FNLin(X)=10^(X/20)         ! Function to compute linear value from dB's.
360                                 !
370  END
```

Line 50 places the analyzer under front panel control allowing the operator to tune the analyzer to position the signal on screen. The span must be chosen such that the signal of interest is the largest response on the screen. When ready, the operator presses CONTinue. The program determines the present span and compares it to 100 kHz, choosing the smaller value. Then, a sweep is taken in single sweep mode, and peak search places the marker on the largest signal, i.e., the fundamental. Marker track is invoked to perform an Auto-Zoom to the span selected above. The signal is then moved to the reference level, the center frequency step size is set to the fundamental frequency, and the amplitude and frequency are output to the controller.

In line 210, the center frequency is incremented once to place the second harmonic on screen. Peak search locates the response and the marker amplitude is output. The same procedure is performed on the third harmonic in line 230.

In line 250, the percent distortion is computed as the root sum of the squares normalized to the fundamental amplitude. As linear values are required in this calculation, a function has been defined in line 350 which converts the dBm values to linear values. The results are finally printed according to the output formats in lines 290—330*.

A typical harmonic distortion measurement might yield the following output:

```
Fund -10.1 dBm    100.0004 MHz
2nd  -41.6 dBm
3rd  -50.8 dBm

Harmonic Distortion =   2.82%
```

*A discussion of PRINT and IMAGE statements can be found in the System 35/45 Operating and Programming Manual.

# 8568A PROGRAMMING CODE LIST

## FRONT PANEL COMMANDS

| | |
|---|---|
| AT | Input attenuation |
| * A1 | Clear-write trace A |
| A2 | Max Hold trace A |
| A3 | Store and view trace A |
| A4 | Store and blank trace A |
| BL | B − DL → B |
| B1 | Clear-write trace B |
| B2 | Max hold trace B |
| B3 | Store and view trace B |
| * B4 | Store and blank trace B |
| * CA | Coupled input attenuation |
| CF | Center frequency |
| * CR | Coupled resolution BW |
| * CS | Coupled step size |
| * CT | Coupled sweep time |
| * CV | Coupled video BW |
| * C1 | A − B off |
| C2 | A − B → A |
| DB | dB |
| DL | Display line |
| DM | dBm |
| DN | Step down |
| DT | Label terminator |
| EE | Enable number entry |
| EK | Enable DATA knob |
| * EM | Erase trace C memory |
| EX | Exchange A and B |
| E1 | Peak Search |
| E2 | Enter marker into center frequency |
| E3 | Enter marker/Δ frequency → step size |
| E4 | Enter marker amplitude → reference level |
| * FA | Start frequency |
| * FB | Stop frequency |
| FS | 0 - 1.5 GHz span |
| GZ | GHz |
| * HD | Hold |
| HZ | Hz |
| IP | Instrument preset |
| I1 | Left RF input |
| * I2 | Right RF input |
| KS | Shift front panel keys |
| * KSA | Amplitude in dBm |
| KSB | Amplitude in dBmV |
| KSC | Amplitude in dBuV |
| KSD | Amplitude in voltage |
| KSE | Title |
| KSF | Measure sweep time |
| KSG | Video averaging on |
| * KSH | Video averaging off |
| KSI | Extended reference level range |
| KSJ | Manual DAC control |
| KSK | Count pilot IF at marker |
| KSL | Noise level off |
| * KSM | Noise level on |
| KSN | Count VTO at marker |
| KSO | Enter Δ → span |
| KSP | Set HP-IB address |
| KSQ | Count signal IF |
| KSR | Diagnostics on |
| * KSS | Second LO auto |
| KST | Second LO down |
| KSU | Second LO up |
| KSV | Frequency offset |
| KSW | Error correction routine |
| KSX | Use correction data |
| KSY | Do not use correction data |
| KSZ | Amplitude offset |
| * KSa | Normal detection |
| KSb | Positive peak detection |
| KSc | A + B → A |
| KSd | Negative peak detection |
| KSe | Sample detection |
| KSf | Power on in last state |
| KSg | CRT beam off |
| KSh | CRT beam on |
| KSi | Exchange B and C |
| KSj | View trace C |
| KSk | Blank trace C |
| KSl | Trace B → trace C |
| KSm | Graticule blanked |
| * KSn | Graticule on |
| KSo | Characters blanked |
| * KSp | Characters on |
| KSq | Step gain off |
| KSr | Service request 102 |
| KSt | Continue sweep from marker |
| KSu | Stop at marker, single sweep |
| KSv | Inhibit phase lock |
| KSw | Display correction data |
| KSx | normal EXT trigger |
| KSy | normal VID trigger |
| KSz | Display storage address |
| KS, | Mixer level |
| KS − | Negative entry |
| KS = | Counter resolution |
| KS( | Save registers locked |
| KS) | Save registers unlocked |
| KS > | Preamp gain, input 2 |
| KS < | Preamp gain, input 1 |
| KS\| | Display storage write |
| KZ | kHz |
| * LG | Enter log scale |
| LN | Linear scale |
| L0 | Display line off |
| * MC0 | Marker frequency count off |
| MC1 | Marker frequency count on |
| MS | msec |
| MV | mV |
| * MT0 | Marker signal track off |
| MT1 | Marker signal track on |
| MZ | MHz |
| * M1 | Marker off |
| M2 | Marker normal |
| M3 | Marker Δ |
| M4 | Marker zoom |
| RB | Resolution BW |
| RC | Recall |
| RL | Reference level |
| SC | sec |
| SP | Frequency span |
| SS | Center frequency step size |
| ST | Sweep time |
| SV | Save |
| * S1 | Sweep continuous |
| S2 | Sweep single |
| TH | Enter threshold |
| * T0 | Threshold off |
| * T1 | Trigger free run |
| T2 | Trigger line |
| T3 | Trigger external |
| T4 | Trigger video |
| UP | Step up |
| UR | Upper right |
| US | μsec |
| UV | μV |
| VB | Video BW |
| 0 to 9 | 0 to 9 |
| . | Decimal point or period |

## OUTPUT COMMANDS

| | |
|---|---|
| DR | Read display and increment address |
| EE | Enable number entry |
| KS123$_{10}$ | Output up to 1001 words |
| LL | Lower left recorder output |
| MA | Marker amplitude output |
| MF | Marker frequency output |
| OA | Output active function |
| OL | Output learn string |
| OT | Output display text |
| O1 | Output format ASCII display units |
| O2 | Output format two 8 bit binary bytes |
| * O3 | Output format ASCII parameter or instrument units |
| O4 | Output format one 8 bit binary byte |
| TA | Output trace A |
| TB | Output trace B |
| UR | Upper right recorder output |

## DISPLAY INPUT COMMANDS

| | |
|---|---|
| * DA | Display address |
| DD | Display write |
| DW | Write into display and increment address |
| * D1 | Display size normal |
| D2 | Display size full CRT |
| D3 | Display size expand |
| GR | Graph |
| IB | Input trace B, binary |
| KS125$_{10}$ | Input up to 1001 display memory words |
| LB | Label |
| PA | Plot absolute |
| * PD | Pen down |
| PR | Plot relative |
| PS | Skip to next display page |
| PU | Pen up |
| SW | Skip to next control instruction |
| TS | Take sweep |

## SERVICE REQUEST COMMANDS

| | |
|---|---|
| R1 | Allow only SRQ 140 |
| R2 | Allow SRQ 140 and 104 |
| * R3 | Allow SRQ 140 and 110 |
| R4 | Allow SRQ 140 and 102 |

| **SRQ** | **Command** | **Bit** | **Definition** |
|---|---|---|---|
| 102 | R4 | 1 | units key pressed |
| 104 | R2 | 2 | end of sweep |
| 110 | R3 | 3 | hardware broken |
| 140 | all | 5 | illegal command |
| 1xx | — | 6 | universal HP-IB service |

* selected with instrument preset

# 8566A PROGRAMMING CODE LIST

## FRONT PANEL COMMANDS

| | |
|---|---|
| AT | Input attenuation |
| * A1 | Clear-write trace A |
| A2 | Max Hold trace A |
| A3 | Store and view trace A |
| A4 | Store and blank trace A |
| BL | B − DL → B |
| B1 | Clear-write trace B |
| B2 | Max hold trace B |
| B3 | Store and view trace B |
| * B4 | Store and blank trace B |
| * CA | Coupled input attenuation |
| CF | Center frequency |
| * CR | Coupled resolution BW |
| * CS | Coupled step size |
| * CT | Coupled sweep time |
| * CV | Coupled video BW |
| * C1 | A − B off |
| C2 | A − B → A |
| DB | dB |
| DL | Display line |
| DM | dBm |
| DN | Step down |
| DT | Label terminator |
| EE | Enable number entry |
| EK | Enable DATA knob |
| * EM | Erase trace C memory |
| EX | Exchange A and B |
| E1 | Peak Search |
| E2 | Enter marker into center frequency |
| E3 | Enter marker/Δ frequency → step size |
| E4 | Enter marker amplitude → reference level |
| * FA | Start frequency |
| * FB | Stop frequency |
| GZ | GHz |
| * HD | Hold |
| HZ | Hz |
| IP | Instrument preset |
| KS | Shift front panel keys |
| * KSA | Amplitude in dBm |
| KSB | Amplitude in dBmV |
| KSC | Amplitude in dBuV |
| KSD | Amplitude in voltage |
| KSE | Title |
| KSF | Measure sweep time |
| KSG | Video averaging on |
| * KSH | Video averaging off |
| KSI | Extended reference level range |
| KSJ | DAC control |
| KSK | Marker to next peak |
| KSL | Noise level off |
| * KSM | Noise level on |
| KSN | Marker to minimum |
| KSO | Enter Δ → span |
| KSP | Set HP-IB address |
| KSQ | Band unlock |
| KSR | Diagnostics on |
| KSS | Fast HP-IB |
| KST | Fast preset 2 — 22 GHz |
| KSU | External mixer preset |
| KSV | Frequency offset |
| KSW | Error correction routine |
| KSX | Use correction data |
| KSY | Do not use correction data |
| KSZ | Amplitude offset |
| * KSa | Normal detection |
| KSb | Positive peak detection |
| KSc | A + B → A |
| KSd | Negative peak detection |
| KSe | Sample detection |
| KSf | Power on in last state |
| KSg | CRT beam off |
| KSh | CRT beam on |
| KSi | Exchange B and C |
| KSj | View trace C |
| KSk | Blank trace C |
| KSl | Trace B → trace C |
| KSm | Graticule blanked |
| * KSn | Graticule on |
| KSo | Characters blanked |
| * KSp | Characters on |
| KSq | Step gain off |
| KSr | Service request 102 |
| KSt | Band lock |
| KSu | Stop at marker, single sweep |
| KSv | Signal identifier ext mixer |
| KSw | Display correction data |
| KSx | normal EXT trigger |
| KSy | normal VID trigger |
| KSz | Display storage address |
| KS. | Mixer level |
| KS= | Factory preselector setting |
| KS − | Negative entry |
| KS( | Save registers locked |
| KS) | Save registers unlocked |
| KS | Display storage write |
| KS# | Turns off YTX self-heating correction |
| KS/ | Manual Preselector peak |
| KS<92> | Enter DL, TH, M2, M3 in display units |
| KZ | kHz |
| LF | Preset 0 — 2.5 GHz |
| * LG | Enter log scale |
| LL | Lower Left |
| LN | Linear scale |
| LO | Display line off |
| MS | msec |
| MV | mV |
| * MT0 | Marker signal track off |
| MT1 | Marker signal track on |
| MZ | MHz |
| * M1 | Marker off |
| M2 | Marker normal |
| M3 | Marker Δ |
| M4 | Marker zoom |
| PP | Preselector peak |
| RB | Resolution BW |
| RC | Recall |
| RL | Reference level |
| SC | sec |
| SP | Frequency span |
| SS | CF step size |
| ST | Sweep time |
| SV | Save |
| * S1 | Sweep continuous |
| S2 | Sweep single |
| TH | Enter threshold |
| * T0 | Threshold off |
| * T1 | Trigger free run |
| T2 | Trigger line |
| T3 | Trigger external |
| T4 | Trigger video |
| UP | Step up |
| UR | Upper right |
| US | μsec |
| UV | μV |
| VB | Video BW |
| 0 to 9 | 0 to 9 |
| . | Decimal point or period |

## OUTPUT COMMANDS

| | |
|---|---|
| DR | Read display and increment address |
| EE | Enable number entry |
| KS<91> | Output amplitude error |
| KS<94> | Output code for harmonic number in binary |
| KS<123> | Display read binary |
| KS<126> | Output every nth value of trace |
| LL | Lower left recorder output |
| MA | Marker amplitude output |
| MF | Marker frequency output |
| OA | Output active function |
| OL | Output learn string |
| OT | Output display text |
| O1 | Output format ASCII display units |
| O2 | Output format two 8 bit binary bytes |
| * O3 | Output format ASCII parameter or instrument units |
| O4 | Output format one 8 bit binary byte |
| TA | Output trace A |
| TB | Output trace B |
| UR | Upper right recorder output |

## DISPLAY INPUT COMMANDS

| | |
|---|---|
| * DA | Display address |
| DD | Display write |
| DW | Write into display and increment address |
| * D1 | Display size normal |
| D2 | Display size full CRT |
| D3 | Display size expand |
| GR | Graph |
| IB | Input trace B, binary |
| KS<39> | Fast Binary DA . . DW |
| KS<125> | Display write binary |
| KS<127> | Display write binary |
| LB | Label |
| PA | Plot absolute |
| * PD | Pen down |
| PR | Plot relative |
| PS | Skip to next display page |
| PU | Pen up |
| SW | Skip to next control instruction |
| TS | Take sweep |

## SERVICE REQUEST COMMANDS

| | |
|---|---|
| R1 | Allow only SRQ 140 |
| R2 | Allow SRQ 140 and 104 |
| * R3 | Allow SRQ 140 and 110 |
| R4 | Allow SRQ 140 and 102 |
| KS<43> | Allow SRQ 140 and 102 |

| SRQ | Command | Bit | Definition |
|---|---|---|---|
| 102 | R4 | 1 | units key pressed |
| 102 | KS<43> | 1 | frequency limit exceeded |
| 104 | R2 | 2 | end of sweep |
| 110 | R3 | 3 | hardware broken |
| 140 | all | 5 | illegal command |
| 1xx | — | 6 | universal HP-IB service |

* selected with instrument preset

For more information, call your local HP Sales Office or nearest Regional Office: **Eastern** (201) 265-5000; **Midwestern** (312) 255-9800; **Southern** (404) 955-1500; **Western** (213) 970-7500; **Canadian** (416) 678-9430. Ask the operator for instrument sales. Or write Hewlett-Packard, 1501 Page Mill Road, Palo Alto, CA 94304. **In Europe**: Hewlett-Packard S.A., 7, rue du Bois-du-Lan, P.O. Box, CH 1217 Meyrin 2, Geneva, Switzerland. **In Japan**: Yokogawa-Hewlett-Packard Ltd., 29-21, Takaido-Higashi 3-chome, Suginami-ku, Tokyo 168.

**5952-9356** Printed in U.S.A.